

Prof. L. Thiele

---

## Technische Informatik 1 - HS 2011

---

### Lösungsvorschläge für Übung 5

Datum: 10.11.2011

---

#### 1 Bit-Test

Zum überprüfen des Status eines I/O- Devices muss oft ein einziges Bit geprüft werden. Implementieren Sie sowohl in C als auch in MIPS-Assembler dafür eine Funktion `int bittest(int x, int n)`. `bittest` soll 1 zurückgeben, falls das Bit an der Position `n` in `x` gesetzt ist, andernfalls 0.

Hinweis 1: Die MIPS Procedure Call Convention, die schon in der Vorlesung bzw. früheren Übungen besprochen worden ist, können Sie dem Unterkapitel A.6 in "Patterson, Hennessy: Computer Organization and Design" entnehmen (online verfügbar: [http://www.cs.wisc.edu/~larus/HP\\_AppA.pdf](http://www.cs.wisc.edu/~larus/HP_AppA.pdf)).

Hinweis 2: Zur Implementierung der Funktion können Sie die folgenden Instruktionen verwenden:

`sllv` (shift left logical variable):      `sllv $rd, $rs, $rt`    `$rd = $rs << $rt`

`srlv` (shift right logical variable):      `srlv $rd, $rs, $rt`    `$rd = $rs >> $rt`

#### Lösung:

Ein entsprechendes C bzw. Assembler Programm könnte wie eine der beiden folgenden Alternativen aussehen:

---

```
int bittest(int x, int n)
{
    if (x & (1 << n))
        return 1;
    return 0;
}
```

```
int bittest(int x, int n) {
    return (x >> n) & 1;
}
```

---

```
bittest:
    li    $t0, 1
    sllv $t0, $t0, $a1
    and  $v0, $a0, $t0
    bgtz $v0, setone
    jr   $ra
setone:
    li    $v0, 1
    jr   $ra
```

```
bittest:
    srlv $t0, $a0, $a1
    andi $v0, $t0, 1
    jr   $ra
```

---

#### 2 Polling eines Memory-Mapped I/O Devices

In dieser Übung wird die I/O-Verarbeitung mit einem MIPS Prozessor betrachtet. Dazu wird der SPIM (SPIM = MIPS rückwärts gelesen) Simulator benutzt, der neben dem CPU Kern auch *memory-mapped*

I/O und *Interrupt-Verarbeitung* simuliert. Der SPIM Simulator verarbeitet ausschliesslich Assembler Code und hat eine einfache, grafische Benutzeroberfläche, die mit dem Kommando `xspim` gestartet wird.

SPIM simuliert zwei memory-mapped Devices: Das Eingabe-Device ermöglicht die Eingabe von Zeichen über die Tastatur. Das Ausgabe-Device ermöglicht die Anzeige von Zeichen auf dem Bildschirm. Prinzipiell sind die beiden Devices voneinander unabhängig. Im Folgenden wird kurz die Funktion der beiden Devices erläutert, siehe auch Abbildung 1.



Abbildung 1: Register zur Steuerung der I/O-Devices.

**Eingabe-Device.** Das Eingabe-Device wird über zwei Register gesteuert, die in den Speicherbereich des MIPS Prozessor eingeblenket sind (daher der Name memory-mapped Device). Das Register *receiver control* ist an der Speicheradresse `0xFFFF0000` eingeblenket. Für diese Aufgabe ist nur das Bit *receiver ready* des Registers *receiver control* relevant: Das Bit *receiver ready* wechselt von 0 auf 1, sobald ein Zeichen auf der Tastatur eingegeben wurde. Bei jedem Lesezugriff auf das Register *receiver data* wechselt das Bit *receiver ready* von 1 auf 0. Im Register *receiver data*, das an der Speicheradresse `0xFFFF0004` eingeblenket ist, wird der ASCII-Wert des zuletzt eingegebenen Zeichens gespeichert.

**Ausgabe-Device.** Das Ausgabe-Device wird ebenfalls über zwei Register gesteuert. Das Register *transmitter control* ist an der Speicheradresse `0xFFFF0008` eingeblenket. Für diese Aufgabe ist nur das Bit *transmitter ready* des Registers *transmitter control* relevant: Das Bit *transmitter ready* ist dann 1, wenn das Ausgabe-Device bereit ist, ein Zeichen auf dem Bildschirm anzuzeigen. Wenn das Bit *transmitter ready* 1 ist, kann in das Register *transmitter data*, das an der Speicheradresse `0xFFFF000C` liegt, der ASCII-Wert eines Zeichens geschrieben werden, das als nächstes angezeigt werden soll. Ein Schreibzugriff auf das Register *transmitter data* hat zur Folge, dass das Bit *transmitter ready* auf 0 gesetzt wird.

## 2.1 Aufgaben

In dieser Aufgabe soll ein Programm erstellt werden, das die Echo-Funktion implementiert. Unter *Echo* versteht man, dass ein vom Benutzer eingegebenes Zeichen am Ausgabe-Device angezeigt wird.

Es soll Polling verwendet werden, um die Echo-Funktion in SPIM zu implementieren. Das folgende Pseudo-Programm zeigt grob die Implementierung der Echo-Funktion mittels Polling:

---

```
while (1) {
    wait_until_character_received();
    x = read_character();
    wait_until_transmitter_ready();
    write_character(x);
}
```

---

- a) Zeichnen Sie ein Zustandsdiagramm, das die Echo-Funktion modelliert. Verwenden Sie die Bits *receiver ready* und *transmitter ready* zur Definition von Ereignissen, die Zustandsübergänge auslösen.

**Lösung:**

Abbildung 2 zeigt das Zustandsdiagramm der Echo-Funktion.




Abbildung 2: Zustandsdiagramm der Echo-Funktion.

- b) Geben Sie ein C Programmfragment an, das ein Zeichen vom Receiver-Device liest. Implementieren Sie dazu in C die beiden Funktionen `wait_until_character_received()` und `read_character()`. Verwenden Sie in `wait_until_character_received()` Polling, um auf die Verfügbarkeit eines Zeichens zu warten.

**Lösung:**

Ein entsprechendes C Fragment könnte wie folgt aussehen:

---

```
volatile int *reg_read_ctrl = (int *)0xFFFF0000;
volatile int *reg_read_data = (int *)0xFFFF0004;
int status;
int character;

while(1) {
    status = *reg_read_ctrl;
    if ((status & 0x1) == 0x1) {
        break;
    }
}
character = *reg_read_data;
```

---

- c) Erstellen Sie ein Assembler Programm, das die Echo-Funktion wie im oben gezeigten Pseudo-Programm implementiert. Im Verzeichnis "polling" finden Sie dazu ein entsprechendes Template `polling.s`:

---

```
.text
.globl main
main:
    addi $sp, $sp, -4    # save $ra.
    sw   $ra, 0($sp)
    lui  $t0, 0xffff    # $t0 = base address for I/O register
                        # use lw $t1, offset ($t0) to load the word at memory address
                        # (0xffff0000 + offset) into register $t1

loop:
    # ***** YOUR CODE GOES HERE *****
    j    loop           # endless loop
```

---

Um das Programm mit SPIM zu simulieren, muss die memory-mapped I/O Funktion explizit aktiviert werden. Starten Sie dazu `xspim` mit der entsprechenden Kommandozeilenoption:  
`xspim -mapped_io polling.s.`

## Lösung:

Ein entsprechendes Assembler Programm könnte wie folgt aussehen:

---

```
.text
.globl main

main:
    addi $sp, $sp, -4    # save $ra.
    sw   $ra, 0($sp)
    lui  $t0, 0xffff    # $t0 = base addr for I/O register

loop:
poll_rx:
    lw   $t1, 0($t0)    # read receiver control reg
    andi $t1, $t1, 1    # mask bit 0 (ready)
    beqz $t1, poll_rx
    lw   $t2, 4($t0)    # read data from receiver

poll_tx:
    lw   $t3, 8($t0)    # read transmitter control reg
    andi $t3, $t3, 1    # mask bit 0 (ready)
    beqz $t3, poll_tx
    sw   $t2, 12($t0)   # write data to transmitter
    j    loop           # endless loop
```

---

### 3 Interrupts (Zusatzaufgabe)

Verwenden Sie nun Interrupts, um die Zeichen vom Receiver Device zu empfangen. Lesen Sie zunächst dazu die Unterkapitel A.7 und A.8 aus "Patterson, Hennessy: Computer Organization and Design" (online verfügbar: [http://www.cs.wisc.edu/~larus/HP\\_AppA.pdf](http://www.cs.wisc.edu/~larus/HP_AppA.pdf)).

- Zunächst müssen Sie die CPU und das I/O-Device entsprechend konfigurieren. Wie aktiviert man die Interrupts der I/O-Devices? Welche Hardware Interrupt-Levels werden dabei verwendet? Schreiben Sie ein Assembler Programm, das Interrupts für das Eingabe-Terminal aktiviert.
- Implementieren Sie nun den Interrupt Handler als Assembler Programm. Im Interrupt Handler soll das gerade eingegebene Zeichen im Ausgabe-Terminal angezeigt werden. Verwenden Sie für die Ausgabe Polling (wie in der letzten Aufgabe). Benutzen Sie das folgende Template:

---

```
.ktext 0x80000180 # forces interrupt routine to be located at 0x80000180
interrupt:
.set noat          # tell assembler to stop using assembly temporary ($at)
sw $at, save_at   # now we can safely use it
.set at           # give back $at to the assembler

# ***** YOUR CODE GOES HERE *****
# save registers (use words in kernel data section for this purpose)
# implement interrupt handler
# restore registers

.set noat
lw $at, save_at
.set at
eret              # exception return

.kdata
save_at: .word 0
save_t0: .word 0
save_t1: .word 0
save_t2: .word 0
save_t3: .word 0
```

---

- Verwenden Sie die Assembler Programme aus a) und b), um das zur Verfügung gestellte Template `interrupt.s` zu vervollständigen. Damit Sie die Unterbrechung des Hauptprogramms durch den Interrupt Handler verfolgen können, gibt das Hauptprogramm kontinuierlich Punkte aus.

Verwenden Sie bei der Simulation mit SPIM folgende Kommandozeilenoptionen, um memory-mapped I/O zu aktivieren und den standardmässigen Interrupt/Exception Handler durch Ihr eigenes Programm zu ersetzen: `xspim -noexception -mapped_io interrupt.s`

#### Lösung:

- Die Interrupts der I/O-Devices werden aktiviert, indem man in den Control Registern das Interrupt Enable Bit setzt. Zudem müssen die Hardware Interrupts auch in der CPU selbst (Coprozessor 0) aktiviert werden. Der Receiver verwendet den HW Interrupt Level 1, der Transmitter den HW Interrupt Level 0. Ein entsprechendes Assembler Programm könnte daher wie folgt aussehen:

---

```
li $t4, 2
sw $t4, 0($t0) # set bit 1 of receiver control register
li $t4, 0x0801 # create bitmask for interrupt status
mtc0 $t4, $12 # write status to status reg of coproc
```

---

b) Ein entsprechendes Assembler Programm könnte wie folgt aussehen:

---

```
.ktext 0x80000180 # forces interrupt routine to be located at 0x80000180
interrupt:
.set noat          # tell assembler to stop using assembly temporary ($at)
sw $at, save_at   # now we can safely use it
.set at           # give back $at to the assembler

sw $t3, save_t3   # save registers. (this is an interrupt routine,
sw $t2, save_t2   # so it has to save the $t registers.)
sw $t1, save_t1
sw $t0, save_t0

lui $t0, 0xffff   # $t0 points to base address of I/O register
lw $t1, 4($t0)    # $t1 = received data

poll:
lw $t2, 8($t0)    # read transmitter control register to $t2
andi $t2, $t2, 1  # mask bit 0 (ready)
beqz $t2, poll    # wait until transmitter ready
sw $t1, 12($t0)

done:
mtc0 $0, $13     # clear cause register
lw $t0, save_t0  # restore registers
lw $t1, save_t1
lw $t2, save_t2
lw $t3, save_t3
.set noat
lw $at, save_at
.set at
eret             # exception return

.kdata
save_at: .word 0
save_t0: .word 0
save_t1: .word 0
save_t2: .word 0
save_t3: .word 0
```

---

c) Ein entsprechendes Assembler Programm könnte wie folgt aussehen:

---

```
#####
# main program
#####
# Main runs in an infinite loop, which continuously outputs '.'
# characters, and starts a new line every 32 characters.
#####

.ktext
.globl main
main:
addi $sp, $sp, -4 # save $ra.
sw $ra, 0($sp)

lui $t0, 0xffff # t0 = base addr for I/O register
li $t1, 32      # initialize character counter
```

```

li    $t2, 46      # load $t2 with '.' (ASCII 46)

# setup interrupts #####
# (a) enable interrupt for receive register
# (b) enable interrupts for CPU, enable HW interrupts level 1

# **** YOUR CODE GOES HERE ****
# simply insert code from subtask a)

loop:
    addiu $t1, $t1, -1 # decrement character counter

poll_tx:
    lw    $t3, 8($t0) # read transmitter control reg
    andi  $t3, $t3, 1 # mask bit 0 (ready)
    beqz  $t3, poll_tx
    sw    $t2, 12($t0) # write data to transmitter
    li    $t2, 46      # load $t2 with '.' (ASCII 46)
    bgtz  $t1, loop   # if character counter > 0 jmp loop
    li    $t1, 32     # reset character counter
    li    $t2, 10     # load $t2 with newline (ASCII 10)
    j     loop        # endless loop

#####
# Interrupt Handler
#####

.ktext 0x80000180 # Forces interrupt routine below to be
                # located at address 0x80000180.

interrupt:
    # save registers
    # read received data
    # transmit received data (polling for transmitter ready)
    # reset interrupt
    # restore registers

    # **** YOUR CODE GOES HERE ****
    # simply insert code from subtask b)

#####
# startup code
# Invoke the routine "main" with arguments: main(argc, argv, envp)
#####
.ktext
.globl __start
__start:
    lw    $a0, 0($sp) # argc
    addiu $a1, $sp, 4 # argv
    addiu $a2, $a1, 4 # envp
    sll  $v0, $a0, 2
    addu $a2, $a2, $v0
    jal  main
    nop
    li   $v0, 10
    syscall          # syscall 10 (exit)

```